Lezione 10 (parte terza)

Enrico Bertolazzi

```
> # costruizione delle fornule di Adams-Muolton-Bashfort
  # con differenze finite
> restart;
```

Verifica della formula di interpolazione

```
> # totale punti di interpolazione
  N := 3 ;

  # successione dei nodi x a partire da x[k+1]=h
  X := [seq(h-i*h,i=0..N)] ;

  # successione dei nodi f(x,y) a partire da f(x[k+1],y[k+1])
  Y := [seq(f[k+1-i],i=0..N)] ;
```

$$N := 3 \tag{1.1}$$

$$X := [h, 0, -h, -2\,h]$$

$$Y := [f_{k+1}, f_k, f_{k-1}, f_{k-2}]$$

```
> # polinomio intepolante
  collect(expand(interp(X,Y,z)),Y) ;
```

$$\left(\frac{1}{6}\,\frac{z^3}{h^3} + \frac{\frac{1}{2}\,z^2}{h^2} + \frac{\frac{1}{3}\,z}{h}\right) f_{k+1} + \left(-\frac{z^2}{h^2} - \frac{1}{2}\,\frac{z^3}{h^3} + \frac{\frac{1}{2}\,z}{h} + 1\right) f_k + \left(\frac{1}{2}\,\frac{z^2}{h^2} - \frac{z}{h} + \frac{\frac{1}{2}\,z^3}{h^3}\right) f_{k-1} + \left(-\frac{1}{6}\,\frac{z^3}{h^3} + \frac{\frac{1}{6}\,z}{h}\right) f_{k-2} \tag{1.2}$$

```
> omega := (z,p) -> mul(z-(1-j)*h,j=0..p-1) ;
  omega(z,0) ;
  omega(z,1) ;
  omega(z,2) ;
  omega(z,3) ;
```

$$\omega := (z, p) \rightarrow mul(z - (1 - j)\,h, j = 0..p - 1) \tag{1.3}$$

$$1$$

$$z - h$$

$$(z - h)\,z$$

$$(z - h)\,z\,(z + h)$$

```
> # polinomio intepolante con le differenze finite
```

```
Delta[0] := f[k+1];
Delta[1] := f[k+1]-f[k] ;
Delta[2] := f[k+1]-2*f[k]+f[k-1] ;
Delta[3] := f[k+1]-3*f[k]+3*f[k-1]-f[k-2] ;

# polinomio interpolante
Q := Delta[0]*omega(z,0)/(0!*h^0)+
     Delta[1]*omega(z,1)/(1!*h^1)+
     Delta[2]*omega(z,2)/(2!*h^2)+
     Delta[3]*omega(z,3)/(3!*h^3):

# stampa del polinomio
collect(expand(simplify(Q)),Y) ;
```

$$\Delta_0 := f_{k+1} \quad (1.4)$$

$$\Delta_1 := f_{k+1} - f_k$$

$$\Delta_2 := f_{k-1} - 2 f_k + f_{k+1}$$

$$\Delta_3 := -f_{k-2} + f_{k+1} - 3 f_k + 3 f_{k-1}$$

$$\left(\frac{1}{6}\frac{z^3}{h^3} + \frac{\frac{1}{2} z^2}{h^2} + \frac{\frac{1}{3} z}{h}\right) f_{k+1} + \left(-\frac{z^2}{h^2} - \frac{1}{2}\frac{z^3}{h^3} + \frac{\frac{1}{2} z}{h} + 1\right) f_k + \left(\frac{1}{2}\frac{z^2}{h^2} - \frac{z}{h} + \frac{\frac{1}{2} z^3}{h^3}\right) f_{k-1} + \left(-\frac{1}{6}\frac{z^3}{h^3} + \frac{\frac{1}{6} z}{h}\right) f_{k-2}$$

Costruzione delle formule di Adams-Bashforth

```
> cfab := p -> int(mul(z+j*h,j=0..p-1)/(h^(p+1) * p!),z=0..h) ;
```

$$cfab := p \rightarrow \int_0^h \frac{mul(z + j\,h, j = 0..p-1)}{h^{p+1}\,p!}\,dz \quad (2.1)$$

```
> seq(cfab(i), i=0..10) ;
```

$$1, \frac{1}{2}, \frac{5}{12}, \frac{3}{8}, \frac{251}{720}, \frac{95}{288}, \frac{19087}{60480}, \frac{5257}{17280}, \frac{1070017}{3628800}, \frac{25713}{89600}, \frac{26842253}{95800320} \quad (2.2)$$

Costruzione delle formule di Adams-Moulton

```
> cfam := p -> int(omega(z,p)/(h^(p+1) * p!),z=0..h) ;
```

$$cfam := p \rightarrow \int_0^h \frac{\omega(z, p)}{h^{p+1}\,p!}\,dz \quad (3.1)$$

```
> seq(cfam(i), i=0..10) ;
```

$$1, -\frac{1}{2}, -\frac{1}{12}, -\frac{1}{24}, -\frac{19}{720}, -\frac{3}{160}, -\frac{863}{60480}, -\frac{275}{24192}, -\frac{33953}{3628800}, -\frac{8183}{1036800}, -\frac{3250433}{479001600} \tag{3.2}$$